

תיכנות ב:
APPLE

נה השבת



12345

משהו על חברת "מחשבת" – מערכות למידה

זו חברה, אשר אחד מתחומי התמחותה העיקריים הוא פיתוח מערכות לימוד עצמי.

מחברי יחידות לימוד אלה עוסקים במשך שנים בפיתוח מערכות ללימוד עצמי עבור המגזר האזרחי והצבאי.

חל איסור מוחלט להעתיק או לשכפל יחידת לימוד זו בשלמותה או בחלקה לכל מטרה שהיא או לעשות בה שימוש מסחרי כלשהו, ללא רשות בכתב מאת חברת מחשבת – מ.ל. בע"מ.

© 1986 – כל הזכויות שמורות למחשבת – מ.ל. בע"מ.

ת.ד. 48032 ת"א מיקוד 61480.

הוצאה לאור – מחשבת מ.ל. בע"מ.

דפוס: דפוס וכריכת מאיר

יהודה הלוי 4, תל-אביב, טלפון: 03-650455

Copyright © 1986 by MAKH-SHEVET M.L. Ltd.

P.O.Box 48032 Tel-Aviv 61480, Israel

APPLE
IIc, IIE, II, II+

BUG

MICROCOMPUTERS BOOKS & SOFTWARE

המרכז לספרות עזר ותוכנה למחשבים

מחשבת

**צעדים נוספים בתכנות
בלימוד עצמי**

יחידה 2

שעונים דיגיטלים ופירמידות

BASIC-2

תוכן העניינים

עמוד

5	פתח דבר.....
6	פרק א - משתנים מספריים.....
17	פרק ב - לולאות סופיות.....
24	פרק ג - בניית שעון דיגיטלי.....
28	פרק ד - שטיחים לפי הזמנה.....
38	פרק ה - שכפול רובוטים.....
44	פרק ו - אנימציה - יצירת תנועה על המסך.....
48	פרק ז - מסגרות לפי מידה.....
51	פרק ח - בניית צורות.....
57	רשימת מושגים נלמדים וסיכום.....
58	תשובות.....

פתח דבר

זוכר את השורה בעזרתה קיבלת פסק זמן?

FOR T=1 TO 1000 : NEXT

שורה זאת איפשרה לרובוט הממוחשב "איציק" לבצע התעמלות בקצב סביר.

ביחידת לימוד זאת "נפצח" ונבין את המסתתר מאחורי הפקודות המופיעות בשורה זאת ונראה כי יש להם מקום מרכזי בכל תוכנית בשפת ה-BASIC.

שמנו לעצמנו כמטרה ללמד ביחידת לימוד זו את המושגים הבסיסיים והחשובים ביותר בשפת הבייסיק:

- משתנים מספריים ומחרוזתיים.

- לולאות סופיות (FOR-NEXT)

נראה כיצד מושגים אלה יפתחו בפנינו עולם ומלואו:
החל בבנית שעון דיגיטלי, וכלה בבנית "פירמידות".

באיחולי המשך

לימוד מהנה ומרתק

דני קדם ויצחק קליסקי
"מחשבת" - מערכות למידה

פרק א

משתנים מספריים

העתק למחשב את התוכנית הבאה:

```
10 HOME
```

```
20 LET N=1
```

```
30 HTAB 20 : VTAB 10 : PRINT N
```

הרץ את התוכנית. על המסך הופיע המספר 1 (1, 10).

מה עליך לעשות בשורה 20 על-מנת שהמחשב ידפיס 2 במרכז המסך?

נסה!

משימתנו תהיה לשנות את התוכנית כך שהמחשב ידפיס מספרים מ-1 והלאה, במרכז המסך. (כל מספר מוחק את קודמו). כדי לבצע משימה זאת עליך לדעת מהו משתנה וכיצד להשתמש בו: (אל תמחק את התוכנית שנמצאת עכשיו בזכרונו של המחשב).

פקוד על המחשב:

```
LET A=6
```

```
LET B=4
```

(אל תשכח ללחוץ RETURN לאחר כל פקודה)

? מה יקרה עכשיו, אם תפקוד על המחשב:

```
PRINT A+B
```

תחילה חשוב - ואחר-כך נסה! (אל תשכח כמוכח RETURN...)

? המחשב הדפיס את המספר : 10 (0, 6, 4, 10)

כיצד קרה הדבר?

ובכן, כאשר פקדת על המחשב:

LET A=6

הוא ביצע את הפעולה המעניינת הבאה:

המחשב הציב באחד מתאי

הזכרון שלו את המספר 6:

וקרא לתא זה בשם: A




דבר דומה קרה כאשר פקדת עליו LET B=4:

הוא הציב בתא זכרון נוסף

את המספר 4 וסימן את התא

בשם: B



ועכשיו, כאשר פקדת עליו:

PRINT A+B

הוא ביצע את הפעולות הבאות:

- חיפש תא בשם A, מצא אותו, והעתיק את המספר הנמצא בתוכו (6) אל מרכז החישובים הנמצא "במוחו".
- המחשב המשיך וחיפש תא בשם B, מצא אותו והעתיק את המספר שהיה בו אל אותו מרכז חישובים.
- מרכז חישובים זה נקרא בשפה מקצועית: היחידה האריתמטית. ביחידה האריתמטית המחשב ביצע את החיבור של 6 ו-4 והציג את התוצאה על המסך.

? האם לאחר שהמחשב ביצע את פעולת החיבור עדיין נמצאים

המספרים 6 ו-4 בתאים A ו-B? כן
בדוק זאת על המחשב:

_____ פקוד עליו להדפיס את ערכו של A (PRINT A).

? ומה לגבי ערכו של B?

מסקנה כדי לבצע פעולות ב-A וב-B העתיק המחשב את ערכם
מן התאים שלהם. הערך של A ו-B נשאר בזכרון!

? מה יתקבל על המסך אם תפקוד על המחשב:

PRINT A*B = 24
(להזכיר: * - הוא סימן הכפל במחשב).

_____ פקוד על המחשב לבצע את פעולת הכפל ובדוק אם צדקת. כן

_____ עכשיו פקוד על המחשב:

LET A=15

? מה ידפיס המחשב לאחר שתפקוד עליו PRINT A: 15
(5, 6, 15, 15 ו-6 ביחד)

_____ פקוד עליו להדפיס את A! האם צדקת? כן

הסבר

- כאשר פקדת קודם LET A=6, הכניס המחשב לתא זכרון את המספר 6 וקרא לתא - A.


- כאשר פקדת אח"כ LET A=15, חיפש המחשב בזכרונו לראות

אם קיים תא בשם A. הוא,

כמובן, מצא תא כזה, ואז

הכניס פנימה את המספר

15, ומחק את המספר 6:



נוכחת לדעת כי A יכול לייצג מספרים כלשהם בהתאם למה שנמצא בתא הזכרון A.

מסיבה זו, "יצור" כמו A בקרא בשם **משתנה**.
ליתר דיוק - משתנה מספרי. (בהמשך היחידה תכיר משתנה נוסף הנקרא - משתנה מחרוזתי.)

____ פקוד על המחשב לתת למשתנה B את הערך 5.

? מה ידפיס המחשב כאשר תפקוד עליו:

PRINT A/B

PRINT 3*A

____ נסה!

? מה יעשה המחשב כאשר תפקוד עליו:

LET C = A + B

במקרה זה הגדרנו משתנה חדש, C, אשר שווה לסכום שני המשתנים A ו-B. (בדוק זאת ע"י PRINT C)

? בדוק מה ידפיס המחשב במקרה הבא:

PRINT D

הסבר

המחשב בודק אם קיים בזכרוננו משתנה בשם D. כמובן שהוא לא מצא משתנה בשם זה ולכן D קיבל את הערך 0.

? מה יקרה כאשר תפקוד:

PRINT A/F

? מדוע מודיע לך המחשב כי ניסית לחלק באפס? (רמז תמצא ביחידה 1, עמוד 15)

קיצורי דרך בכתיבה

עד כה הגדרנו משתנים בעזרת פקודת LET. המחשב שלך יודע לעשות זאת גם אם תוותר על כתיבת המילה LET:

$A=50$

נסה לכתוב:

בקש מן המחשב להדפיס את ערכו של A.

שמות של משתנים

איזה שם מותר לתת למשתנה?

השם שהמחשב נותן לתא בזכרון בו נמצא הערך של המשתנה, חייב להתחיל באות אנגלית. אחרי אות זו יכולות לבוא עוד אותיות אנגליות או ספרות, כרצונך.

? אילו מבין שמות המשתנים הבאים אינו חוקי? :

APPLE, SUM, B, 2D, A1B2, 25, PRINT

(תשובה 1 בעמוד 58)

כתוב במחשב: $1X=2$

(כדי לבדוק אם 1X יכול להיות שם של משתנה).

נסה כעת להדפיס את ערך המשתנה (PRINT 1X).

המחשב מודיע לך כי זו שגיאה.

אמור למחשב להדפיס את התוכנית. (אינך זוכר איך? - פנה

ליחידה 1 עמוד 28)

איזו שורה נוספה עכשיו לתוכנית שכבר נמצאת בזכרוננו של המחשב?

אתה רואה כי המחשב פשוט התיחס ל-1 כאל מספר בתוכנית!

מחק את שורה 1 מהתוכנית!

(אינך זוכר איך? - ראה יחידה 1 עמוד 31)

נסכם

- שמו של משתנה חייב להתחיל באות אנגלית.
- אחרי אות זו יכולות לבוא אותיות נוספות או ספרות (כרצונך).

אגב: מותר לך לתת למשתנה שם ארוך יותר:

כתוב במחשב: $PGIOT = 3$ (פגיעות = 3)

אך המחשב יתייחס רק לשתי האותיות הראשונות:

הדפס: `PRINT PG`

`PRINT PGIOT`

`PRINT PG13`

השם הכתוב על תא הזכרון של משתנה זה הוא PG (PGIOT, PG)

אתגר מחשבתי

הגדר במחשב משתנה חדש, X , שיהיה שווה ל-10:

$X = 10$

(להזכירך: ניתן לכתוב $X=10$ במקום `LET X=10` - וכמוכן

`RETURN`)

? מה יהיה ערכו של X לאחר שתפקוד על המחשב:

`LET X = X+1`

כדי לדעת אם צדקת, פקוד על המחשב לבצע פקודה זו (אפשר גם

בלי `LET`) ואח"כ פקוד עליו להדפיס את X .

קיבלת כי X שווה ל 11 (1, 10, 11).

כיצד התקבלה תוצאה זו?

הפקודה $X=X+1$ אין פירושה ש- $10=11$!

הפקודה $X=X+1$ פירושה עבור המחשב הוראה לביצוע סדרה של פעולות:

- המחשב מבצע תחילה את מה שכתוב בצד ימין של השוויון:

במקרה שלנו לוקח את הערך הנוכחי של המשתנה X ומוסיף

לו 1: $(10+1)$.

- את התוצאה המחשב מציב במשתנה ששמו רשום בצד שמאל של

השוויון: במקרה שלנו מופיע שם שוב המשתנה X ולכן

הוא מקבל עכשיו את הערך 11.

משימה

ועכשיו, לאחר שהבנת את מושג המשתנה, תוכל לפצח את המשימה

שהטלנו עליך קודם לכן.

הצג את התוכנית, שכבר נמצאת בזכרונו של המחשב, על המסך.

(זאת התוכנית שכבר כתבת בעמוד 6)

עליך לשנות את התוכנית כך שהמחשב ידפיס במרכז המסך מספרים מ-1 והלאה. כלומר: ידפיס 1 במרכז המסך, מיד לאחר מכן ידפיס באותו מקום 2 (ה-1 נמחק, כמובן), עליו 3 וכן הלאה.

בדרך כל הכלים התכנותיים לכתוב את התוכנית - נסה בכוחות עצמך.

מתקשה?

רמז 1: עליך להוסיף עוד שתי שורות לתוכנית.
(ואולי רמז זה יספיק לך?)

רמז 2: השלם את החסר:

$$40 \quad \underline{N} = \underline{N} + 1$$

50 GO TO 30

אגב האם המספרים רצים מהר מדי על המסך?

• עצור את ריצת התוכנית (אינך זוכר איך? - ראה יחידה 1 עמוד 37).

• הכנס "פסק-זמן" במקום המתאים בתוכנית, בדומה לזה שהכנסת כדי להאט את קצב ההתעמלות של אי-ציק ביחידה 1 עמוד 53.
(תשובה 2 בעמוד 58)

התבונן במסך: האם זה מזכיר לך התחלפות מספרים בשעון דיגיטלי? - בפרק ג תבנה בעצמך שעון כזה.

נלך צעד צעד ונראה כיצד התוכנית עובדת:

שורה 20: אחרי ביצוע שורה זאת המשתנה N שווה ל: 3,2,1

שורה 30: המחשב מדפיס את ערכו של N.

שורה 40: N מקבל ערך חדש השווה לערך הקודם בתוספת 1:

$$\begin{array}{c} N = N + 1 \\ \uparrow \quad \uparrow \\ \text{קודם} \quad \text{חדש} \end{array}$$

אחרי ביצוע שורה זו בפעם הראשונה, N מקבל את

הערך: (1, 2, 3)


שורה 50: פקודה זו כבר מוכרת לך מיחידה 1. הפקודה מחזירה

את המחשב לשורה 30 בתוכנית.

וכך הלאה...

הייה אתה המחשב

כדי לעקוב אחר הגלגולים שעובר המשתנה N בזמן ריצת התוכנית, הייה אתה המחשב - עבור שורה שורה וכתוב את ערכו של המשתנה



התוכנית שבמחשב עדיין רצה על המסך שלך?

אם לא תעצור את המחשב - האם הוא ימשיך לספור עד אינסוף?

כמעט! המחשב יכול לספור עד המספר העצום 10^{38} !

(הספרה 1 עם "זנב" של 38 אפסים)

תן למחשב לחשב ולהדפיס את התוצאה של:

$$1,000,000 * 1,000,000$$

אתה רואה כי בתגובה המחשב כתב:

$$1E+12$$

$$1E+12 = 1 * 10^{12}$$

שפירושו:

מהו המספר הגדול ביותר שהמחשב עדיין רושם עם אפסים?

(תשובה 4 בעמוד 13)

משימות

בצע בתוכנית שבמחשב את השינויים הבאים:

• שנה את התוכנית כך שיודפסו רק המספרים האי-זוגיים:

1, 3, 5, 7, ...

(תשובה 5 בעמוד 58)

• בצע שינוי כזה שיודפסו המספרים: 5, 10, 15, ...

? מה יקרה אם תשנה את שורה 50 ל:

50 GO TO 20

פינת המשחק: "בול 1000"

בעזרת מה שלמדת זה עתה תוכל לשחק משחק קטן עם עצמך או עם חבריך:

כתוב את התוכנית הבאה:

```
5 HOME
10 HTAB 18 : VTAB 10 : PRINT "BOOM"
20 B=B+1
30 GO TO 20
```

הרץ את התוכנית.

? האם המחשב סופר על המסך? כן (כן/לא)

דע לך כי מוחו של המחשב סופר ללא הרף במהירות רבה!
(המחשב מתרוצץ ללא הפסקה בין שורה 20 לבין שורה 30)
רוצה להשתכנע שהוא סופר?

עצור את ריצתו ופקוד עליו להדפיס את ערכו של המשתנה B.
(אינך זוכר כיצד? ראה עמוד 7)

המספר שהודפס הוא ערכו של B כשהתוכנית נעצרה.

חזור והרץ את התוכנית. עצור את המחשב והצג שוב את ערכו של B. מה ערכו של B הפעם?

1000

וכעת למשחק עצמו:

מטרת המשחק: לעצור את המחשב קרוב ככל האפשר למספר 1000.
שיטת המשחק: כל אחד בתורו מקבל 5 נסיונות. בכל ניסיון הוא מריץ את התוכנית ועוצר אותה כאשר הוא חושב ש-B הגיע ל-1000.
כל אחד זוכר את השיא שלו. (המספר הקרוב ביותר ל-1000).
מי שהגיע למספר הקרוב ביותר ל-1000 - ניצח!
סיימת לשחק?

מחק את התוכנית מהזכרון ועבור הלאה...

משימות לסיום הפרק

עכשיו, לאחר שאתה מכיר את מושג המשתנה, נטיל עליך כמה משימות גרפיות:

כתוב תוכנית שתדפיס על המסך אותיות A במשבצות הבאות:

	1	2	3	4	5	...
1						
2				A		
3						
4						
5				A		
6						
7						
8				A		
9						
10						
11				A		
12				:		
:				:		
:				:		

ה-A יודפסו בעמודה 5

ובשורות 2, 5, 8, ...

עד שורה 20.

קרא למשתנה בו אתה משתמש

בשם V.

(הרשת המלאה של המסך מופיעה

ביחידה 1 עמוד 71)

מתקשה?

ניתן לך רמז:

התבונן בשלוש השורות הבאות:

10 HOME

20 LET V=2

30 HTAB 5 : VTAB V : PRINT "A"

? מה עושות שלוש שורות אלה?

העתק והרץ אותן במחשב.

ועכשיו הוסף עוד שתי שורות כך שכל ה-A יודפסו.

(תשובה 6 בעמוד 58)

אגב, המחשב מודיע לך: 30 ILLEGAL QUANTITY ERROR IN

הסיבה לכך היא, שההדפסה "גלשה" מתוך המסך, וזאת מהסיבה

שהמשתנה V קיבל ערך כזה שלא נמצא במסך.

רוצה להשתכנע?

פקוד על המחשב להדפיס את ערכו של המשתנה V.

אם פעלת נכון - המחשב הדפיס: 26.

ובאמת, כאשר אתה תפקוד:

HTAB 5 : VTAB 26 : PRINT "A"

המחשב "יגלוש" מעבר למסך ושם הוא בודאי לא יכול להדפיס,

ולכן המחשב מודיע לך על שגיאה!

(זכור כי במסך יש רק 24 שורות).

עכשיו שנה את התוכנית כך שבמקום כל ה-A יופיעו המספרים:

2, 5, 8, 11, 14...

(תשובה 7 בעמוד 58)

ולסיום

? מה יקרה אם תשנה את שורה 30:

30 HTAB V : VTAB V : PRINT "0"

חשוב היטב לפני שתריץ את התוכנית במחשב. לאחר שהחלטת מה

תעשה התוכנית, העתק והרץ אותה.

? האם התוכנית ביצעה את אשר חשבת שהיא תעשה?

(אם לא - פנה להסבר נוסף בתשובה 8 בעמוד 58)

? ומה יקרה אם במקום "0" תבקש להדפיס את האות O בלי מרכאות?
בדוק עצמך על המסך.

? ומה יקרה אם במקום האות O תכניס בתוכנית את V? (ללא מרכאות)

לאחר שתחשוב בדוק עצמך על המסך.

פרק ב

לולאות סופיות

עכשיו, לאחר שהכרת את מושג המשתנה, נוכל לפנות ולפצח את הפקודות:

```
FOR T=1 TO 1000 : NEXT
```

שהופיעו בפסק-זמן בתוכנית בה אי-ציק ביצע התעמלות.

לולאות FOR-NEXT

האם אתה זוכר את התוכנית:

```
10 PRINT "שמך"
```

```
20 GO TO 10
```

מיחידה 1?

הפקודה GO TO המופיעה בשורה 20 גרמה לכך שהמחשב הסתובב בלולאה אינסופית וביצע פעם אחר פעם את שורה 10. פעולה זו נפסקה רק כאשר לחצת על המקשים CTRL-C.


בעזרת ההוראות FOR-NEXT תוכל להכתיב למחשב כמה פעמים לבצע כל הוראה:

האם תוכל לנחש כמה פעמים ירשום המחשב את שמך בתוכנית הבאה? **?**

```
10 FOR N=1 TO 5
```

```
20 PRINT "שמך"
```

```
30 NEXT N
```

העתק את התוכנית למחשב והרץ אותה! 
(לפני כן נקה את הזכרון מכל תוכנית קודמת וכן את המסך.)

אתה רואה כי שמך הופיע 5 פעמים זה מתחת לזה.

שנה את התוכנית כך, שהמחשב ישכפל את שמך 8 פעמים.

(תשובה 9 בעמוד 59)

נסכים:

כאשר אתה כותב תוכנית עם צמד הפקודות FOR-NEXT אתה יכול לשלוט במספר הפעמים שהמחשב מבצע את השורות הכתובות בין צמד פקודות זה.

למשל:


```
10 FOR N=1 TO 20  
[ 20 print "michael"  
  30 .....  
  40 .....  
50 NEXT N
```

כמה פעמים יבצע המחשב את שורות 20, 30 ו-40?

(תשובה 10 בעמוד 59)

משימה

א. כתוב תוכנית שתרשום שעה A לאורך אלכסון במרכז המסך:
(השתמש ב-FOR-NEXT)



(תשובה 11 בעמוד 59)

(רמז: החבונן בתוכנית
שבעמוד 16)

ב. כתוב תוכנית שתרשום את שמך 5 פעמים בצורה כזאת על המסך (שוב, השתמש ב-FOR-NEXT)



(תשובה 12 בעמוד 59)

משתנה לולאה

מחק את התוכנית ועבור הלואה:

? מה תבצע התוכנית הזאת:

```
5 X=1
10 FOR N=1 TO 20
20 PRINT X
30 X=X+1
40 NEXT N
50 PRINT "SOF" (סוף = SOF)
```

לאחר שחשבת - העתק את התוכנית והרץ אותה.

למען האמת ניתן לקצר את התוכנית שכתבנו זה עתה ולכתוב

אותה כך:

```
10 FOR N=1 TO 20
20 PRINT N
30 NEXT N
40 PRINT "SOF"
```


שים לב לדבר מעניין:

המשתנה N המופיע בשורה 20, מופיע גם בשורה 10 ובשורה 30!

כדי לעזור לך להבין את אשר קורה למשתנה N נעקוב אחר

ביצוע התוכנית שורה אחר שורה:


RUN



? בדוק עצמך אם הבנת: הייה אתה המחשב וכתוב את ערכי המשתנה N במקומות הריקים:

RUN

↓
10 FOR N=1 TO 3 ---
20 PRINT N -----
30 NEXT N -----
40 PRINT "SOF"



החלט גם, מתי אתה כמחשב מדפיס "סוף".

(תשובה 13 בעמוד 59)

נחזור לתוכנית הנמצאת בזכרונו של המחשב:

? מה יקרה אם נשנה את שורה 10 ל:

10 FOR N=5 TO 20

התוכנית תספור מ 5 עד 20.
נסה וראה אם צדקת.

? מה יקרה אם תמחק לגמרי את שורה 30?

מחק את שורה 30 מן התוכנית והרץ.

המספר היחיד שהמחשב רשם הוא 5 (5, 10, 15, 20)

נוכחת לדעת כי ללא שורה 30 (NEXT) המחשב אינו חוזר כלל לשורה 10 (FOR ...) - הוא מדפיס את הערך הראשון בלבד של המשתנה N (במקרה זה - 5).

מחק את התוכנית מן הזכרון

? מה תעשה התוכנית הבאה:

10 FOR N=0 TO 5

20 PRINT N*N

30 NEXT N

חשוב לפני שתבדוק עצמך על המחשב!

משימה

זוכר את המשימה אשר בה ביקשנו ממך לשרטט שבע פעמים A
באלכסון? (ראה עמוד 18)
עכשיו, כאשר אתה יודע על קיומו של משתנה הלולאה, חוכל
לקצר תוכנית זאת. נסה!
(תשובה 14 בעמוד 59)

לולאת השהייה

מה קורה כאשר אין כלל שורות בין צמד הפקודות FOR-NEXT
היוצרות את הלולאה:
10 PRINT "START" (START באנגלית - התחלה)
20 FOR T=1 TO 1000
30 NEXT T
40 PRINT "END" (END באנגלית - סוף)

מה עושה תוכנית זו? מה תפקידן של שורות 20 ו-30?
לאחר שחשבת...

נקה את הזכרון והכנס את התוכנית הזאת.

מה עליך לעשות כדי להגדיל פי שניים את הזמן בין הופעת
ה-START על המסך לבין ה-END?
(תשובה 15 בעמוד 59)

צמד השורות 20 ו-30 יצר השהייה.
כאשר המחשב מגיע לשורות 20 ו-30 הוא נמצא בלולאה, והוא
מבצע אותה $\frac{1}{1000}$ (1000, 100, 1) פעמים. (המשתנה N מתחיל
מ-1 ורק כאשר N גדול מ-1000 המחשב יוצא מן הלולאה.)
הזמן שבו ביצע המחשב 1000 סיבובים בלולאה יוצר השהייה.
מובן, לכן, שכלל שהמחשב מבצע יותר סיבובים בלולאה -
ההשהייה ארוכה יותר.
למען האמת זוהי בדיוק הדרך שבה השגנו השהייה (פסק-זמן)
בתוכנית שבה אי-ציק ביצע התעמלות בחוברת מספר 1.
אתה ודאי זוכר את ההשהייה בצורה הבאה:

71

2. ולאחר מכן חוזר לפקודה 1 וכך חוזר 1000 פעמים!

הרץ את התוכנית וראה כי מתקבלת אותה תוצאה.

(האם זכרת למחוק את שורה 30?)

התוכנית העייפה

התבונן בתוכנית הבאה:

```
10 PRINT X;
```

```
20 FOR T=1 TO X * X : NEXT T
```

30 $X=X+1$ כפול

40 GO TO 10

? מדוע קראנו לתוכנית זו "התוכנית העייפה"?

הרץ את התוכנית ובדוק בעצמך.

(אינר מביו? פנה לתשובה 16 בעמוד 60)

משימה


כתוב תוכנית שתביע את שמך

באלכסון על המסך:

השם רץ מהר מדי? הכנס השהייה!

רמז: הזכר איך אי-ציק ביצע

התעמלות בחוברת 1.




(תשובה 17 בעמוד 60)

נסכים:

• השורה 50:
  10
:
50 GO TO 10

יוצרת לולאה אינסופית המחשב חוזר ומבצע את הפקודות בין השורות 10 ו-50 פעם אחר פעם ללא הגבלה.

• השורות 10 ו-50:

 10 FOR N=1 TO 10
:
50 NEXT N

יוצרות לולאה סופית המחשב חוזר ומבצע את הפקודות כמספר הפעמים שרשום בשורה 10.

• המשתנה N בלולאה הסופית נקרא משתנה לולאה.

בהמשך יחידת הלימוד נעשה שימוש נרחב במושגים אלה!

פרק ג

בנית שעון דיגיטלי

מבוא

עכשיו, לאחר שהבנת את פקודות הלולאה FOR-NEXT, אתה יכול לבנות במחשב שלך שעון דיגיטלי. אנו מתארים לעצמנו כי מצוי בביתך שעון דיגיטלי כלשהו. רצוי מאוד ששעון כזה יהיה לידך בזמן שאתה בונה שעון דיגיטלי במחשב.

ספירת השניות

תחילה נתרכז בבניית הצגת השניות של השעון הדיגיטלי:

התבונן בשעון הדיגיטלי שלך, על הספרות המציגות את השניות, וענה: ?

השעון מתחיל לספור את השניות מ- 0 (0, 1). האם מופיע גם המספר 60 בסוף הדקה? כן/לא.

אתה רואה כי השעון מתחיל מ-0 ומגיע עד 59. ואז במקום לרשום 60 הוא חוזר ורשם 0 ומתחיל מן ההתחלה. כדי לגרום למחשב להדפיס שניות על המסך נבנה את התוכנית בשלבים:

כתוב תוכנית שתדפיס, במרכז המסך בערך, מספרים המתחלפים מ-0 עד 59. 1

המספרים המופיעים על המסך מתחלפים במהירות גדולה מאוד.

הכנס לולאת השהייה לתוך התוכנית.

(תשובה 18 בעמוד 60)

בשלב שני הוסף פקודה לתוכנית שתגרום למחשב לחזור ולספור מ-0 כל פעם שהוא מגיע ל-59.

(תשובה 19 בעמוד 60)

הרץ את התוכנית.

?

האם אתה מבחין ב"בג" שיש בתוכנית? /
בכל פעם שהמחשב מתחיל לספור מחדש מ-0, הסיפרה 9 מהמחזור הקודם עדיין נשארת על המסך ומתקבל:
9 0

שנה את התוכנית כך שתגבר על הבג.

(תשובה 20 בעמוד 60)

הכנסת השעון לקצב שניות אמיתי

?

האם קצב הופעת המספרים על המסך (מהיר יותר, איטי יותר)
מזה של השעון הדיגיטלי?

?

איזו השהייה עליך להכניס על מנת להתאימו לקצב של שניות?

FOR T=1 TO 64 : NEXT T

בצע כמה ניסויים בעזרת שעון דיגיטלי עד אשר תגיע לקצב הקרוב ביותר לזה של השניות בשעון הדיגיטלי.

(תשובה 21 בעמוד 60)

נפנה לספירת הדקות

התכוון בשעון דיגיטלי: מה קורה כאשר השעון הדיגיטלי משלים לספור דקה שלמה?

במקרה זה סיפרת הדקות קופצת ב-1.

?

עליך להוסיף לתוכנית פקודות מתאימות כך, שהשעון יספור גם דקות.

(רמז: עליך להוסיף לולאת FOR-NEXT מתאימה לתוכנית.)

הקפד על כך שהתוכנית שלך תבצע את הדברים הבאים:

• ברגע הפעלת השעון יראה המסך כך:


\emptyset	:	\emptyset
↑		↑
דקות		שניות

(תשובה 22 בעמוד 0)

שתילת לולאות אחת בתוך השניה


בוא נעצור לרגע ונתבונן בתוכנית שלנו:

בתוכנית ישנן 3 לולאות FOR-NEXT:



אתה רואה כי ניתן להשתמש במספר רב של לולאות בתוכנית "השתולות" האחת בתוך השניה.

האם המחשב מסוגל לבצע תוכנית שבה יש לולאות FOR-NEXT מצטלבות כמו זו: ?



(תשובה 23 בעמוד 61)

קיצור כתיבה

עד כה נהגנו לכתוב את שם המשתנה אחרי NEXT:

NEXT S

למען האמת, המחשב שלך מוכן לקבל את פקודת ה-NEXT גם ללא שם המשתנה:

FOR T=1 TO 1050 : NEXT

האם אין סכנה שהמחשב יתבלבל:

FOR D =

FOR S =

:

NEXT → NEXT של מה?

NEXT → NEXT של מה?

אל תדאג!

כבר אמרנו שאסור לבנות לולאות מצטלבות. ולכן, גם אם אינך כותב את שם המשתנה בפקודת NEXT, המחשב יודע למה התכוונת:

המחשב משך באופן אוטומטי את ה-FOR וה-NEXT הפנימיים ביותר וכך הלאה...

FOR D =
FOR S =
:
NEXT
NEXT

המלצה: בכל זאת - אל תתעצל וכתוב את שם המשתנה בפקודת NEXT במקומות בהם אתה עלול להתבלבל.

ולסיום

אם עדיין לא נמאס לך השעון - הוסף לו גם שעות!
(ואולי גם תאריך?)

פרק ד

שטיחים לפי הזמנה

משימות שכפול

נתחיל את הפרק בכמה משימות שכפול:
כתוב תוכנית שתשכפל את הסימן "\$" 10 פעמים על המסך
במשבצות הבאות:

	1	2	3	4	5	6	7	8	9	11	12	...
1												
2			\$	\$	\$	\$	\$	\$	\$	\$	\$	
3												
:												
:												
:												

בתוכנית לא צריך
להופיע יותר מסימן
\$ אחד.

זקוק לרמז?

השלם את התוכנית הבאה:

```
5 HOME
10 FOR H=___ TO ___
20 HTAB___ : VTAB___ : PRINT "$"
30 _____
```

(תשובה 24 בעמוד

אם התוכנית רצה מהר מדי אולי כדאי שתכניס לולאת שהייה.
הרץ את התוכנית.


	1	2	3	4	5	...
1						
2						
3				\$		
4				\$		
5				\$		
6				\$		
7				\$		
8						
:						
:						

כתוב תוכנית דומה שתשכפל את
עמודת הדולרים הבאה:
(קרא למשתנה שלך: V)

(תשובה 25 בעמוד

שטיח של דולרים:

עכשיו עליך לכתוב תוכנית שתיצור "שטיח" (ריבוע) של דולרים:
זכור:




- בתוכנית מותר רק ל-\$ אחד להופיע!

רמז 1:

אנו רוצים לכתוב תוכנית, אשר תחילה תשכפל את השורה העליונה של הדולרים, ולאחר מכן תעבור לשורה השנייה ותתחיל לרשום אותה מהתחלה. וכך הלאה עד שהיא תשלים ותכתוב את מספר השורות המבוקש.

האיור הבא מדגים כיצד התוכנית מכפילה את הדולרים בשלוש השורות הראשונות:



כיצד תכתוב את התוכנית כך, שכל פעם שהיא משלימה שורה היא תקפוץ, באופן אוטומטי, לשורה הבאה ותתחיל שוב לשכפל דולרים? אם אתה יודע לענות על שאלה זו אולי אתה יכול לכתוב את התוכנית? נסה עכשיו בלי להמשיך לרמז 2.

רמז 2:

בתוכנית הבאה חסרות שתי פקודות:

```
10 HOME
20 ?
30 FOR H=4 TO 13
40 HTAB H : VTAB V : PRINT "$"
50 NEXT H
60 ?
```

אם תשלים אותן נכון תקבל את התוכנית המבוקשת.
(תשובה 26 בעמוד 61)

הכנס בתוכנית שינויים כאלה, שהשטיח יהיה בגודל 5×5 .
שנה את התוכנית כך, שיודפס שטיח של 15 שורות ו-15 עמודים
קרוב לודאי שאתה נאלץ לבצע שינויים בשני מקומות בתוכנית
בכל פעם שאתה רוצה לשנות את גודל הריבוע:

משימה

שנה את התוכנית כך, שכדי לשנות את גודל השטיח, תצטרך
לשנות מספר רק במקום אחד בתוכנית!
(רמז: אתה צריך להשתמש במשתנה נוסף)
התגבר על הבגים לפני שתפנה לתשובה 27 בעמוד 61.

משכללים את התוכנית

אנו מעוניינים שהמחשב ישאל אותך לפני ציור השטיח באיזה
גודל שטיח אתה מעוניין.
לשם כך עליך ללמוד פקודה חדשה:

INPUT (אינפוט) - הכנס לתוכנית:

הוסף את שורה 12:

12 INPUT L

ושנה, בשורות 20 ו-30 את שם המשתנה שקבע את גודל השטיח
ל-L אם הוא לא כזה.

הרץ את התוכנית.

על המסך מופיע סימן שאלה: ?

עכשיו המחשב שואל אותך איזה ערך אתה רוצה שיהיה למשתנה L.
(INPUT L)

הקש את המספר (גודל השטיח) ואל תשכח RETURN.

המחשב שרטט שטיח בדיוק במידה שבחרת!

כיצד קרה הדבר?

כאשר המחשב הגיע לשורה: INPUT L 12 הוא נעצר, הציג על המסך סימן-שאלה, וחיכה שתכניס לו את הערך של המשתנה L. לאחר שהכנסת מספר, ולחצת RETURN הוא שומר בזכרונו כי המשתנה L שווה למספר זה והוא ממשיך בביצוע התוכנית. בכל מקום בתוכנית בו מופיע המשתנה L, הוא שווה למספר שהכנסת ב-INPUT.

משימה

שנה את התוכנית כך, שבסיום של הדפסת כל שטיח, המחשב ישאל אותך, באופן אוטומטי, פעם נוספת מה הגודל בו אתה רוצה את השטיח הבא. כל זאת - בלי למחוק את השטיח הקודם שכבר צויר. כמובן, שלאחר שהכנסת את ערכו של L, דאג לכך שהשטיח הקודם יימחק ובמקומו יודפס שטיח חדש בגודל הרצוי. (ושוב מופיע סימן-שאלה של ה-INPUT על המסך).

(תשובה 28 בעמוד 61)

איך יוצאים מ-INPUT?

נסה לעצור את התוכנית כאשר המחשב נמצא ב-INPUT.

תיווכח לדעת ש: CTRL-C אינו עוצר את התוכנית!

לשם כך:

לחץ CTRL-RESET. כאשר תרפה מן הלחיצה ישמע צפצוף קצר

והתוכנית תיעצר.

שכלול מדהים!

רוצה שהמחשב גם ישאל אותך מאיזה סימן אתה רוצה שהוא יבנה

את השטיח? (במקום \$)

הכנס את השינויים הבאים בתוכנית:

• רשום שורת INPUT נוספת:

14 INPUT A\$

• והחלף בשורה 40 את ה-"\$" ל-A\$.

(ללא מרכאות).

ללאחר הכנסת התוספות הרץ את התוכנית.

סימן שאלה מופיע על המסך: המחשב שואל אותך לגודל השטיח.
בחר בגודל ולחץ RETURN.

עכשיו מופיע עוד סימן-שאלה - המחשב רוצה לדעת מאיזה סימן
לבנות את השטיח. בחר סימן כלשהו ולחץ שוב RETURN - המחשב
הממושמע משרטט שטיח בגודל ובצורה שפקדת עליו!
לפני שנכנס להסבר מפורט - שחק להנאתך עם התוכנית:

• הכנס גדלי-שטיח שונים.

• הכנס בכל פעם סימן אחר.

נהנית? נעבור להסבר יותר מפורט:

משתנה מחרוזת

בשורה 12 יש משתנה L שקלט את המספר שהכנסת:

```
12 INPUT L
```

בשורה 14 (ובשורה 40), לעומת זאת, מופיע משתנה מסוג חדש
הנקרא משתנה מחרוזת:

```
14 INPUT A$
```

מהי מחרוזת?

למען האמת, כבר השתמשת בעבר במחרוזות, מבלי לדעת שהן
"מחרוזות":

מה מבצע המחשב בפקודה הבאה: `PRINT "2+5"`

המחשב אינו מבצע את פעולת החשבון אלא פשוט מעתיק את כל מ
שכתוב בתוך המרכאות.

זכור: כל דבר שתשים בתוך מרכאות יחשב כמחרוזת:

"מחרוזת"

בפקודה הבאה:

```
PRINT "4*7=" ; 4*7
```

איזה חלק הוא מחרוזת ואיזה חלק הוא ביטוי חשבוני?

(תשובה 29 בעמוד

? מכמה סימנים מורכבת כל אחת מהמחרוזות הבאות:

"APPLE"

"*25/7^"

(תשובה 30 בעמוד 61)

? ומה יקרה אם תרצה שהמחשב ידפיס את המחרוזת הבאה

"BASIC"

ובטעות פקדת עליו:

PRINT BASIC

כמובן שהמחשב הבין זאת כשם של משתנה מספרי ורשם את הערך 0.

ומהו משתנה מחרוזת?

כתוב במחשב:

LET B\$ = "XØR" (אל תמחק את התוכנית


שישנה בזכרון)

? מה יקרה אם תפקוד:

PRINT B\$

נסה. כיצד תסביר את התוצאה?

התהליך דומה מאוד למה שקורה במחשב עם משתנים מספריים:



המחשב קרא לאחד

התאים שבזכרון: B\$

והכניס לתוכו את

המחרוזת "XØR"

כאשר פקדת PRINT B\$ המחשב מצא את התא הנקרא B\$, העתיק

מתוכו את המחרוזת והדפיס אותה על המסך.

? מה יקרה, לדעתך, אם תכתוב: LET B\$ = 5

נסה ותיווכח אם צדקת.

המחשב הודיע לך TYPE MISMATCH - כלומר: אין התאמה בין השם

של התא בזכרון (B\$ - משתנה מחרוזתי) לבין מה שניסית להכניס

לתוכו (5 - ערך מספרי).

? האם המחשב יקבל "5" = B\$ LET ? נסה!

נסה להכניס מחרוזת לתא של משתנה מספרי (למשל LET B="5").

? מה תעשה תוכנית שכזאת:

```
500 X$ = "שמך"  
510 FOR N=1 TO 10  
520 PRINT X$  
530 NEXT
```

כדי לבדוק אם אתה צודק אל תמחק את התוכנית הנמצאת כבר בזכרון!

העתק את התוכנית, כתוב RUN 500 ולחץ RETURN.

המחשב יתחיל "לרוץ" משורה 500 ואילך, וכך תוכל לבדוק אם צדקת לגבי התוכנית הקטנה הכתובה למעלה.

סיימת? מחק את השורות מ-500 ואילך.

שמות של משתני מחרוזת

הכללים החלים על שמות של משתנים מספריים נכונים גם לגבי משתני-מחרוזת:

- השם חייב להתחיל באות אנגלית.
- אחרי אות זו יכולות לבוא אותיות וספרות כרצונך.
- המחשב מזהה רק את שני הסימנים הראשונים של השם ומתעלם מן השאר.

אבל:

שם של משתנה מחרוזת חייב להסתיים תמיד בסימן !\$
מתוך השמות הבאים של משתני מחרוזת - איזה חוקיים ואיזה המחשב לא יקבל: ?

A\$, 2A\$, AAA\$, B3, C\$\$, C*\$

(תשובה 31 בעמוד)

המחרוזת הריקה

זוכר מה קרה כאשר ביקשת מהמחשב להדפיס משתנה מספרי שלא הגדרת קודם לכן מה ערכו?
הוא פשוט הדפיס \emptyset .

ואיך יגיב המחשב על משתנה מחרוזת שלא הוגדר?
נחש מה יופיע על המסך כאשר תפקוד על המחשב:

?

`PRINT "A" ; ZX$; "B"`




ד



ג



ב



א

לאחר שניסית לנחש...

כתוב `"B" ; ZX$; "A" PRINT` (ולחץ RETURN).

הסבר

`ZX$` הוא משתנה-מחרוזתי שלא הוגדר עדיין (לא קיבל כל ערך קודם). במקרה כזה - כאשר אתה לא בותן למשתנה מחרוזתי ערך כלשהו - הוא מקבל את הערך של המחרוזת הריקה:

$$ZX\$ = ""$$

מחרוזת ריקה היא מחרוזת שאינה מכילה כלום בין המרכאות - אפילו לא רווח (ובודאי לא \emptyset).

וחזרה לתוכנית שלנו:

בתוכנית שלנו רצינו שהמחשב ישרטט את השטיח עם אחד מהסימנים המופיעים על המקשים - אות, מספר ועוד. לשם כך השתמשנו במשתנה מחרוזת `A$`.

כאשר המחשב הגיע לשורה:

```
14 INPUT A$
```

הוא מציג סימן-שאלה על המסך ומחכה שתגיד לו למה אתה רוצה
שמשתנה המחרוזת A\$ יהיה שווה.

שים לב: אינך צריך, במקרה זה, לרשום מרכאות - המחשב
יודע כי עליו לקלוט מחרוזת. (איך הוא יודע?)

לאחר שהקשת על אחד הסימנים (עם RETURN, כמובן), המחשב
יודע כי משתנה המחרוזת A\$ שווה לסימן שהקשת.
בכל פעם שהוא מגיע לשורה

```
40... : PRINT A$
```

הוא מדפיס את הסימן שתקתקת ל-A\$.
מה יקרה אם משתנה המחרוזת A\$ יהיה שווה למחרוזת בת 2
סימנים? (למשל: A\$ = "AB")
נסה!

כותרות ל-INPUT

לתוכנית שלך יש עכשיו שתי שורות עם פקודות INPUT:

- עבור המשתנה המספרי L.
- עבור המשתנה המחרוזתי A\$.

שמת בודאי לב, כי סימני-השאלה שהמחשב מציג הם זהים.
כדי לא להתבלבל בין שני ה-INPUT אפשר להורות למחשב
להדפיס כותרת מיוחדת לכל סימן-שאלה שהוא מציג, וכך
נדע בדיוק לאיזה נתון הוא מחכה.

הנה דוגמא:

שנה את שורה 12:

```
12 INPUT "L=" ; L
```

המשתנה נשאר L ←
אל תשכח הכותרת היא
מחרוזת

הרץ את התוכנית.

עכשיו מתקבל: $L =$

ואתה יוזע בדיוק לאיזה משתנה המחשב מחכה.

עצור את ריצת התוכנית. (אינך זוכר? ראה עמוד 31)

עכשיו הכנס כותרת גם ל-INPUT השני.

לדוגמא:

```
14 INPUT "SIMAN=" ; A$
```

ולסיום

הצג את התוכנית שפיתחת על המסך.

אתה רואה כי כבר בשלב זה התוכנית שפיתחת היא די מורכבת.

תוך שימוש במושג המשתנה ובמושג הלולאה למדת להשתמש בשני

מושגים חשובים חדשים:

- ה-INPUT

- משתנה המחרוזת.

פרק ה

שכפול דוֹבוּטִים

שמוש ב־STEP

יש לכתוב תוכנית אשר תשכפל 5 דולרים על המסך:

	1	2	3	4	5	6	7	8	9	10	12	14	16	18	...
1															
2															
3															
4															
5			\$				\$			\$			\$		\$
6															
:															
:															

השלם את התוכנית הבאה כדי שהיא תעשה זאת: ☐

```

5 HOME
10 H = _____
20 FOR N=1 TO _____
30 HTAB _____ : VTAB _____ : _____
35 FOR T=1 TO 400 : NEXT T
40 H = _____ + _____
50 NEXT N
    
```

הכנס את התוכנית למחשב והרץ אותה. ☐

(תשובה 32 בעמוד 62)

קיצור התוכנית

ניתן לקצר באופן משמעותי את התוכנית בעזרת ההוראה
STEP (סָטֵפ) שפירושה - צעד.

שנה את התוכנית הנמצאת בזכרון באופן, שתקבל התוכנית:

```

5 HOME
20 FOR H=2 TO 18 STEP 4
30 HTAB H : VTAB 4 : PRINT "$"
35 FOR T=1 TO 400 : NEXT T
40 NEXT H

```

הרץ את התוכנית. האם קיבלת אותה תוצאה כמו בתוכנית הקודמת?

מה יקרה לדעתך אם תשנה את STEP 4 ל-STEP 2?

הכנס את השינוי וראה אם צדקת.

נסכם:

הוספת STEP נותנת הוראה למחשב לשנות את המשתנה המופיע ב-FOR-NEXT "בקפיצות" (צעדים) בהתאם למספר הרשום אחרי STEP:

- STEP 4 אומר למחשב כי על משתנה הלולאה לגדול כל פעם ב-4.
- STEP 2 אומר למחשב כי על משתנה הלולאה לגדול כל פעם ב-2.

משימה

הוסף עוד שתי פקודות לתוכנית כך שהיא תדפיס 3 שורות של \$ ברווחים של שתי שורות ריקות בין שורה לשורה. כך:

	1	2	3	4	5	6	7	8	9	10	12	14	16	18	20	22	24	26	...
1																			
2																			
3			\$				\$				\$			\$			\$		
4																			
5																			
6			\$				\$				\$			\$			\$		
7																			
8																			
9			\$				\$				\$			\$			\$		
10																			
11																			

(תשובה 33 בעמוד 62)

שנה את התוכנית כך שתקבל את השטיח הזה על כל המסך.

מחק את התוכנית ועבור הלאה...

? מה תעשה התוכנית הבאה:

```
10 FOR N=1 TO 40 STEP 5
20 PRINT N
30 NEXT N
```

לאחר שחשבת - הרץ את התוכנית ובדוק עצמך.

? ומה יקרה לדעתך בתוכנית הבאה:

```
10 FOR N=100 TO 0 STEP -10
20 PRINT N
30 NEXT N
```


(שים לב לסימן המינוס)

הרץ אותה במחשב.

הפעם הלך המחשב אחורנית: הוא התחיל ב-100 והגיע עד 0.

משימה

השתמש ב-STEP שלילי כדי לשרטט אלכסון של דולרים על המסך:



זקוק לרמז:

הזכר בתוכנית ששרטטה את האלכסון השני (ראה עמוד 18)

(תשובה 34 בעמוד 62)

בעזרת הוראת ה-STEP נוכל לשכפל צורות כרצוננו...

שכפול דוֹבוּטִים

נא הכר את אי-לי אחיו הצעיר

של הרובוט אי-ציק מהיחידה

הקודמת:

(מיד יופיעו על המסך גם

אחיו התאומים.)

	1	2	3	4	5	...
1						
2			0			
3		#	#	#		
4		X		X		
5						
...						

נתחיל בכתיבת התוכנית המצוירת את אי-לי:

א. השלם את השורות הבאות על-מנת לקבל את ראשו של אי-לי

בדיוק במשבצת הנמצאת בעמודה 3 ובשורה 2:

הערה: הראש בנוי מן

האות "0"

	1	2	3	4	5	6	...
1							
2			0	שורה			
3			עמודה				
4							
5							
...							

5 HOME

10 LET H = _____

20 LET V = _____

30 HTAB H : VTAB V : _____

אנו כותבים את אי-לי באופן כללי, בעזרת המשתנים H ו-V, כדי שנוכל בהמשך לשכפל אותו.

ב. עכשיו צריך להוסיף את הכתפיים והגוף:

שים לב כי הכתפיים מתחילות בשורה אחת _____ (מעל, מתחת)

לראש, ועמודה אחת _____ (שמאלה, ימינה) מהראש.

לכן, הפקודה לקבלת הכתפיים והגוף תהיה:

40 HTAB H- _____ : VTAB V+ _____ : PRINT _____

(שים לב! זה סימן מינוס)

השלם את השורה והכנס אותה למחשב. בדוק אם מתקבלת

התוצאה הרצויה.

אם אתה מתקשה פנה לתשובה 35 בעמוד 62.

אגב: האם H ו-V משנים את ערכם לאחר שהמחשב מבצע את שורה 40?

ג. בעבור לרגליים:

הוסף שורה שתצייר את הרגליים:

50

הרץ ובדוק אם אתה מקבל את אי-לי על המסך.

(תשובה 36 בעמוד 62)

משימה

שנה את התוכנית כך שהמחשב ישאל אותך באיזה עמודה (H) ובאיזה שורה (V) אתה רוצה שיופיע ראשו של אי-לי. ומיד הוא ישרטט, באופן אוטומטי, רובוט נוסף בעמודה H ובשורה V- מבלי למחוק את הרובוט הקודם שכבר הופיע. מיד לאחר שהמחשב משלים את שרטוט הרובוט הנוסף הוא חוזר ושואל אותך למקום חדש של רובוט נוסף... בסה להתגבר על המשימה בכוחות עצמך. אם אתה מתקשה, נאמר לך, כי אחד המושגים שלמדת עליהם בפרק הקודם יבואו לעזרתך.

סיימת לכתוב ולהריץ את התוכנית? -

השווה לתוכנית שבתשובה 37 בעמוד 63.

עצור את ריצת התוכנית כאשר המחשב נמצא ב-INPUT.

(אינך זוכר כיצד? - ראה עמוד 31)

שכפול אוטומטי של אי-לי:

עד כה גרמת אתה למחשב לשכפל את אי-לי, לאחר שהכנסת
ב-INPUT ערכים למשתנה H ולמשתנה V.
האם תוכל לפתח תוכנית שתשכפל באופן אוטומטי מספר שורות
של אי-לי?
נתחיל בשורה אחת:

עליך להוסיף ולשנות את התוכנית כך שהמחשב ידפיס את אי-לי
4 פעמים לרוחב המסך במקומות הבאים:

	1	2	3	4	5	6	7	8	9	10	11	12	13	14	15	16	17
V=2			0					0				0				0	
		#	#	#			#	#	#		#	#	#		#	#	#
		X		X			X		X		X		X		X		X

(רמז: עליך להשתמש ב-FOR-NEXT וב-STEP)

(תשובה 38 בעמוד 43)

ועכשיו הוסף פקודות כך שהתוכנית תשכפל שלוש שורות של אי-לי
זו מתחת לזו.

(חשוב - כיצד קיבלנו את השורות של ה-"\$" בעמוד 39)

(תשובה 39 בעמוד 63)

שנה את התוכנית כך, שכל המסך יתמלא ברובוטים.

(תשובה 40 בעמוד 63)

ולסיום

- שכפל את אי-לי באלכסון.
- שכפל אותו בשני אלכסונים.


פרק 1

אנימציה – יצירת תנועה על המסך

זוכר את אי-ציק עושה התעמלות בוקר ביחידה הראשונה?
עכשיו ננצל את הכלים החדשים שרכשת בינתיים כדי להניע
צורות על המסך.

משימה 1


עליך לכתוב תוכנית אשר תגרום לצורה $>$ "לעוף" לרוחב המסך
מעמודה 2 עד עמודה 38 בשורה 4:
כאשר הצורה $>$ תגיע לעמודה 18
היא תתחיל שוב לנוע משמאל
לימין וכן הלאה.



דע לך כי אתה יודע את כל הפקודות הדרושות.
רוצה לנסות לבד? - נסה עכשיו לכתוב את התוכנית ולהריצה.

אם אתה מתקשה כדאי שתפנה לרמז בעמוד הבא:

הנה מבנה התוכנית: עליך לכתוב תוכנית אשר



האם מתוך מבנה התוכנית תוכל כבר לכתוב את התוכנית בעצמך? - נסה!

הרץ את התוכנית. אם מצאת "בגים" בתוכנית - היה סבלני ונסה לתקן אותם.

הערה "הציפור" טסה מהר מדי?

האט אותה בעזרת לולאת השהייה.

(תשובה 41 בעמוד 63)

משימה 2

עליך לשנות את התוכנית כך שתחילה תראה הצורה - ">" נעה משמאל לימין (יוצאת "מקיר" 1 ומגיעה ל"קיר" 2)

X	>	X
"קיר" 1		"קיר" 2

כאשר הצורה מגיעה ל"קיר" 2 היא תתהפך: "<" ותנוע בחזרה

X	<	X	מ"קיר" 2 ל"קיר" 1
1		2	

- את "הקירות" בנה מהאות "X".
- גם את הצורה ההפוכה "<" תמצא על לוח המקשים.

בבנית התוכנית הקפד:

- "הציפור" אינה מוחקת את "הקירות".
- הציפור עפה הלוך ושוב ללא הפסק.
- הקירות אינם מהבהבים.
- הציפור מגיעה במעופה ממש עד הקירות.

(תשובה 42 בעמוד 63)

פקוד על המחשב לצפצף

ודאי שמעת את המחשב שלך מצפצף מפעם לפעם, ביחוד כשיש
...SYNTAX ERROR

כעת תלמד לשלוט בצפצוף זה ולפקוד על המחשב לצפצף לפי פקודה:
כתוב במחשב:

PRINT "

וכעת לחץ CTRL-G (החזק את מקש CTRL לחוץ ולחץ גם על מקש
האות - G).

נשמע צפצוף קצר - אך דבר לא השתנה על המסך.

המשך ולחץ CTRL-G מספר פעמים.

כעת סגור את המרכאות:

PRINT ""

המחרוזת נראית כמחרוזת ריקה אבל...

לחץ RETURN

להפתעתך הרבה השמיע המחשב מספר צפצופים - כמספר הפעמים
שלחצת CTRL-G לפני שסגרת את המרכאות!

המחרוזת הקולית

בכל פעם שלחצת CTRL-G נכנסה מחרוזת מיוחדת במינה לתוך
המחשבת: "מחרוזת קולית".

כאשר לחצת RETURN ופקדת על המחשב להדפיס את המחרוזת
הקולית - הוא לא הדפיס דבר אלא ביצע את המחרוזת הקולית
והשמיע צליל.

פרק ז

מסגרות לפי מידה

- בפרק זה עליך לפתח תוכנית אשר תבצע את הפעולות הבאות:
- המחשב שואל אותך איזה שם אתה רוצה להכניס למסגרת.
 - אתה כותב את השם - ואז המחשב כותב אותו במרכז המסך (בערך) ומצייר מסביבו מסגרת המתאימה את עצמה באופן אוטומטי לגודל השם שבחרת!

לדוגמא:

```
* * * * *
*           *
*  D A N I  *
*           *
* * * * *
```

ועבור שם ארוך יותר - מסגרת ארוכה:

```
* * * * *
*           *
* K A L I S K Y *
*           *
* * * * *
```

LEN - אורך המחרוזת

כדי לכתוב את התוכנית עליך להכיר את ההוראה LEN:

LEN הוא קיצור של המילה LENGTH שפירושו - אורך.

כדי להבין את פעולת ההוראה LEN:

פקוד על המחשב

```
PRINT LEN ("ABCDE")
```

(אל תשכח את הסוגריים)

לאחר שלחצת RETURN המחשב כתב את המספר : 5 ?

ומספר זה שווה בדיוק למספר הסימנים המופיעים במחרוזת - זהו אורך המחרוזת.

? מה יכתוב המחשב על המסך לאחר ביצוע התוכנית הבאה:

20 X1\$ = "A1B2C3"

30 PRINT LEN (X1\$)

נסה.

נסכים:

פקודת LEN אומרת למחשב לספור את מספר הסימנים שבתוך המחרוזת.

ועכשיו - לאחר שהכרת את פקודת ה-LEN, אתה יכול להשתמש בה על-מנת לבנות את תוכנית המסגרות. אנו מציעים כי תעיז ותנסה להתמודד עם התוכנית לבדך - אפילו אם הדבר יגזול ממך זמן מה. אם אתה בכל זאת מתקשה - נפתח את התוכנית שלב אחר שלב:

שלב 1

• ראשית כל עליך לדעת לכתוב תוכנית המציירת מסגרת מלבנית,


באורך מוגדר וברוחב מוגדר

ללא קשר לשם שיופיע בתוכן:

```
      H
      ↑
V ← * * * * *
    *               *
    *               *
    *               *
    * * * * * * * *
```

אורך המלבן המופיע כאן הוא 8 כוכביות ורוחבו - 5 כוכביות.

• עליך לכתוב תוכנית שתשרטט את המלבן על המסך, כאשר הכוכבית השמאלית העליונה נמצאת בעמודה 10 ($H=10$) ובשורה 5 ($V=5$).



כיצד תעשה זאת?

ישנן מספר דרכים:

יתכן כי תרצה לשרטט תחילה את צלע 1, לאחר מכן 2, 3 ולבסוף 4, או בתחילה את 1 ו-3 ולאחר מכן 2 ו-4. בחר בדרך הנוחה לך.

(תשובה 45 בעמוד 64)

? מה עליך לעשות בתוכנית כך שהמחשב ישאל אותך באיזה אורך מלבן אתה מעוניין והמחשב ישרטט אותו באופן אוטומטי.
(תשובה 46 בעמוד 64)

שלב 2

כעת נעבור לכתיבת השם עצמו.

• עיקר הקושי בתוכנית הוא להתאים את אורך המסגרת לאורך השם:
הנה דוגמא:
השם מורכב מ-4 סימנים.
המסגרת מורכבת מ- $\frac{1}{2}$ (4, 8) סימנים.
אורך המסגרת

```

* * * * *
*           *
*  D A N I  *
*           *
* * * * *

```

לכן: אורך המסגרת = אורך השם + $\frac{1}{2}$ (4, 8, 2)

תוכל להשוות את תוכניתך לזו שבתשובה 47 בעמוד 65.

שכלולים

- המחשב ישאל אותך גם מאיזה סימן לבנות את המסגרת.
- המחשב ידפיס את שמך ב-INVERSE.
- המחשב ידפיס את שמך ב-FLASH.


פרק ח

בנית צורות

עצה ללומד: אם תמצא פרק זה כקשה ל"עיכול", אל תנסה "לבלוע" אותו בשלב זה. קפוץ לעמוד 57 לסיכום ומשם ליחידה 3. בהמשך תוכל לחזור לפרק זה ולהתמודד איתו בשנית וביתר הצלחה.

משימה 1: בנית "פירמידה"

עליך לכתוב תוכנית, שתשרטט את הפירמידה הבאה:



התוכנית צריכה להיות בנויה כך, שהיא תכלול רק פקודת PRINT אחת.

אתה מתקשה? בעזרת התרגילים הבאים תוכל לפענח את סוד בנית הפירמידה:

תרגיל 1

? מה עושה התוכנית הבאה:

10 HOME : INVERSE

20 A=10 : B=20 : V=5

30 FOR H=A TO B

40 HTAB H:VTAB V:PRINT "□"


50 NEXT H

200 NORMAL □ - מציין רווח

אחרי שחשבת - בדוק עצמך במחשב. נסה כעת לכתוב את תוכנית הפירמידה.

תרגיל 2

אם עדיין לא הצלחת - הנה תרגיל עזר נוסף:
עליך לשנות ולהוסיף פקודות לתוכנית שבתרגיל 1 כך,
שהתוכנית תרשום גם את השורה שבתרגיל 1, ותוסיף עוד שורה:



ועכשיו, לאחר שבצעת את שני תרגילי העזר בהצלחה, אתה חייב לדעת כיצד לכתוב את התוכנית של הפירמידה.

ואם בכל זאת לא הצלחת


(תשובה 48 בעמוד 65)

משכללים את הפירמידה

תוכל להכניס את השכלולים הבאים כרצונך:

- המחשב שואל אותך מאיזה סימן לבנות את הפירמידה - בונה אותה מהסימן שבחרת - וחוזר ושואל אותך.
- המחשב משמיע צפצוף בגמר ציור של כל שורה משורות הפירמידה.
- (אינך זוכר כיצד לקבל צפצופים? - פנה לעמוד 46)

מה עליך לעשות בתוכנית כדי לקבל צורה כזאת?




(תשובה 49 בעמוד 65)

האם אתה מעוניין לבנות צורה כזאת?

אנו משאירים אותך להתמודד עם


זה לבדך...



משימה 2: "ריבועים מתנפחים"

עליך יהיה לכתוב תוכנית, אשר תצייר מלבנים זה בתוך זה:
בהתחלה תצייר התוכנית את המלבן הפנימי, לאחר מכן את שאר

המלבנים בזה אחר זה:



ושוב, נסה בכוחות עצמך לכתוב את התוכנית.


עצה מועילה: על-מנת לבנות את התוכנית, עליך לחשוב בתחילה

כיצד לבנות מלבן אחד כללי. אח"כ עליך להוסיף פקודות

שישכפלו את המלבן וינפחו אותו.

אם אתה מתקשה, פנה לתרגילי העזר שבהמשך.

תרגיל עזר



עכשיו גבנה תוכנית לבניית

מלבן כללי בין העמודות A

ו-B לבין השורות C ו-D.

גבנה את התוכנית בשלבים:

שלב א

כתוב תוכנית, בה ישרטט המחשב את צלע 1 משמאל לימין.

תן ערכים ל-A, B ו-C כרצונך.

(תשובה 50 בעמוד 65)

הרץ את התוכנית במחשב.

שלב ב

הוסף לתוכנית פקודות, שבעזרתן ימשיך המחשב וישרטט את צלע 2

מלמעלה למטה. (תן ערך גם ל-D)

(תשובה 51 בעמוד 65)

הרץ את התוכנית.

שלב ג

עכשיו הוסף לתוכנית את הפקודות המתאימות כך, שהיא תשרטט את צלע 3 מימין לשמאל.

הערה: עליך להשתמש ב-STEP 1- (שלילי).

(תשובה 52 בעמוד 66)

הרץ את התוכנית על המחשב.

שלב ד

ועכשיו לא יקשה עליך להשלים ולכתוב את הצלע הרביעית.


כתוב אותה במחשב והרץ את התוכנית.

הפעם לא ניתן לך תשובה. אם לא הצלחת לשרטט את הצלע

הרביעית, סימן שלא הבנת משהו קודם. על כן, חזור והתחל

את תרגיל העזר מהתחלה.

שלב ה



מה צריך לקרות ל-A, B, C ו-D כדי שהתוכנית תצייר 3 מלבנים נוספים?
הכנס את הפקודות המתאימות כך, שתקבל את שלושת המלבנים הנוספים.

(תשובה 53 בעמוד 66)

בוא "נשחק" עם התוכנית


עכשיו, לאחר שהתוכנית הכללית הסתיימה, הוסף לה פקודה כך, שהמחשב ישרטט את המלבנים באופן הבא:



- שנה את גודל המלבנים ומיקומם.
- שנה רק את B והשאר את A, C ו-D קבועים.
- הרץ את התוכנית וראה מה קיבלת.
- וכך המשיך לשחק עם שאר המשתנים.
- הכנס צלילים כרצונך.
- הכנס FLASH ו-INVERSE.

לפניך מספר הצעות לשרטוט צורות במחשב

התכניות הן בהישג ידך. יחד עם זאת דרושה סבלנות רבה לכתוב אותן. אם אתה מעוניין בסה אותן:



שאלת חזרה לסכום

לפניך רשימת מושגים שלמדת ביחידה זו.
עבור על כל אחד מהם ובדוק אם אתה זוכר אותם.
(תוכל לרענן את זכרוןך בעזרת מספרי העמודים המופיעים
בסוגריים ליד כל מושג.):

(30) INPUT	(19) משתנה
(32) מחרוזת	(10) שמות של משתנים
(33) משתנה מחרוזת	(11) LET A=A+1
(34) שמות של משתני מחרוזת	(17) FOR-NEXT
(35) המחרוזת הריקה	(19) משתנה לולאה
(38) STEP	(21) לולאת השהיה
(46) המחרוזת הקולית	(23) לולאה אינסופית
(48) LEN	(26) FOR-NEXT לולאות

שתולות אחת בתוך

השניה

לסיום...

עברת כברת דרך רצינית. אתה כבר יודע לכתוב תוכנית תוך שימוש
ברחב בלולאות FOR-NEXT ומשתנים, וכל זאת תוך ניצול תכונותיו
המיוחדות של מחשב ה-APPLE.

יתכן שבשלב זה אתה כבר יכול להמריא על כבפי דמיוןך.

אם כן, אף אחד לא מונע זאת ממך.

ויתכן שתרצה להמשיך וללמוד את יחידה 3:

"פיתוח משחקי מחשב-טלויזיה", אשר עם סיומה תדע להשתמש

במירב הכלים העומדים לרשותך בשפת ה-BASIC.

ביחידה 1 - עמוד 74 תמצא פרטים נוספים על יחידה 3.

תשובות

תשובה 1 מעמוד 10


השמות הבאים אינם חוקיים: PRINT, 25, 2D

תשובה 2 מעמוד 12

45 FOR T=1 TO 300 : NEXT T

תשובה 3 מעמוד 13

20 N=1
30 PRINT N
40 N=N+1
50 GO TO 30



תשובה 4 מעמוד 13

המספר הגדול ביותר שהמחשב עדיין כותב אותו בצורה הרגילה, עם אפסים הוא: 100,000,000

תשובה 5 מעמוד 13

יש לשנות: 40 N=N+2

תשובה 6 מעמוד 15

יש להוסיף: 40 V=V+3

50 GO TO 30

תשובה 7 מעמוד 16

יש לשנות: 30 HTAB 5 : VTAB V: PRINT N

תשובה 8 מעמוד 16

המחשב מתחיל "לרוץ" ומגיע לשורה 30. בשלב זה $V=2$ ולכן הוא מדפיס את האות "O" בעמודה 2, שורה 2. בשורה 40 V גדל ב-3 והוא כבר שווה ל-5. כאשר המחשב מגיע פעם נוספת לשורה 30 הוא מדפיס את האות "O" בעמודה 5, שורה 5, וכך הלאה. לכן מתקבל אלכסון.

תשובה 9 מעמוד 18

יש לשנות: 10 FOR N=1 TO 8

תשובה 10 מעמוד 18

המחשב יבצע שורות אלה 20 פעמים. (FOR N=1 TO 20)

תשובה 11 מעמוד 18

```
10 HOME
20 M=12
30 FOR N=1 TO 5
40 HTAB M : VTAB M : PRINT "A"
50 M=M+1
60 NEXT N
```

תשובה 12 מעמוד 18

יש לשנות: 30 FOR N=1 TO 5
40 HTAB M : VTAB M : PRINT "DANY"

תשובה 13 מעמוד 20

RUN
↓

10 FOR N=1 TO 3	---	N=1	→	N=2	→	N=3
20 PRINT N	-----	N=1		N=2		N=3
30 NEXT N	-----	N=2	→	N=3	→	N=4
40 PRINT "SOF"					↓	SOF

כש-N מקבל ערך הגדול מ-3 (TO 3) המחשב יוצא מה-FOR-NEXT וממשיך לשורה 40.

תשובה 14 מעמוד 21

```
10 HOME
20 FOR N=12 TO 18
30 HTAB N : VTAB N : PRINT "A"
40 NEXT N
```

תשובה 15 מעמוד 21

יש לשנות: 20 FOR T=1 TO 2000

תשובה 16 מעמוד 22

20 FOR T=1 TO X*X :NEXT T הסוד נעוץ בשורות
30 X=X+1

כיוון ש-X גדל ב-1 בכל פעם, הרי שהמחשב "סופר" בתחילה רק עד 1
(1×1), אח"כ עד 4 (2×2) ואח"כ עד 9 (3×3) וכן הלאה עד מספרים
גדולים מאוד.

תשובה 17 מעמוד 22

10 HOME ✓
20 FOR N=1 TO 23
30 HTAB N:VTAB N:PRINT "DANY"
40 FOR T=1 TO 100 : NEXT
50 HTAB N:VTAB N:PRINT " "
60 NEXT N

תשובה 18 מעמוד 24

10 HOME
20 FOR S=0 TO 59
30 HTAB 25 : VTAB 12 : PRINT S
40 FOR T=1 TO 300 : NEXT
50 NEXT S

תשובה 19 מעמוד 25

60 GO TO 20 יש להוסיף:

תשובה 20 מעמוד 25

30 HTAB 25 : VTAB 12 : PRINT S ; " " יש לשנות:

תשובה 21 מעמוד 25

40 FOR T=1 TO 1050 : NEXT יש לשנות:

תשובה 22 מעמוד 26

15 FOR D=0 TO 59 יש להוסיף או לשנות:
30 HTAB 25:VTAB 12:PRINT D; ":"; S ; " "
55 NEXT D

תשובה 23 מעמוד 26

התשובה היא שלילית! אסור לכתוב בתוכנית לולאות מצטלבות (נסה).

תשובה 24 מעמוד 28

```
10 HOME
20 FOR H=3 TO 12
30 HTAB H:VTAB 2:PRINT "$"
40 NEXT H
```

תשובה 25 מעמוד 28

```
10 HOME
20 FOR V=3 TO 7
30 HTAB 4:VTAB V:PRINT "$"
40 NEXT V
```

תשובה 26 מעמוד 29

```
20 FOR V=4 TO 13
60 NEXT V
```

יש להוסיף:

תשובה 27 מעמוד 30

```
12 L=10
20 FOR V=4 TO L+3
30 FOR H=4 TO L+3
```

יש להוסיף או לשנות:

תשובה 28 מעמוד 31

```
15 HOME
70 GO TO 12
```

יש להוסיף:

תשובה 29 מעמוד 32

4*7 הוא בלטיח חשבוני ו-"4*7=" היא מחרוזת.

תשובה 30 מעמוד 33

המחרוזת: "APPLE" בנויה מ-5 סימנים.
המחרוזת: "+*25/7^" בנויה מ-7 סימנים.

תשובה 31 מעמוד 34

השמות הבאים אינם חוקיים למשתני-מחרוזת:

2A\$, B3, C\$\$, C*\$

תשובה 32 מעמוד 38

```
5 HOME
10 H=3
20 FOR N=1 TO 5
30 HTAB H : VTAB 5 : PRINT "$"
35 FOR T=1 TO 400 : NEXT T
40 H=H+4
50 NEXT N
```

תשובה 33 מעמוד 39

יש להוסיף או לשנות: 10 FOR V=3 TO 9 STEP 3
50 NEXT V

תשובה 34 מעמוד 40

```
10 HOME : V=1
20 FOR H=30 TO 10 STEP -1
30 HTAB H : VTAB V : PRINT "$"
40 V=V+1
50 NEXT H
```

תשובה 35 מעמוד 42

```
5 HOME
10 H=3
20 V=2
30 HTAB H : VTAB V : PRINT "O"
40 HTAB H-1 : VTAB V+1 : PRINT "###"
```

תשובה 36 מעמוד 42

יש להוסיף:

```
50 HTAB H-1 : VTAB V+2 : PRINT "X □ X"
```


תשובה 37 מעמוד 42

10 INPUT "H="; H
20 INPUT "V="; V
60 GO TO 10

יש להוסיף:

תשובה 38 מעמוד 43

10 FOR H=3 TO 18 STEP 5
20 V=2
60 NEXT H

יש לשנות:

תשובה 39 מעמוד 43

20 FOR V=2 TO 10 STEP 4
70 NEXT V

יש לשנות או להוסיף:

תשובה 40 מעמוד 43

10 FOR H=3 TO 38 STEP 5
20 FOR V=2 TO 18 STEP 4

יש לשנות:

תשובה 41 מעמוד 45

10 HOME
20 FOR H=2 TO 38
30 HTAB H : VTAB 4 : PRINT ">"
40 FOR T=1 TO 100 : NEXT T
50 HTAB H : VTAB 4 : PRINT "□"
60 NEXT H

תשובה 42 מעמוד 46

12 HTAB 1:VTAB 4: PRINT "X"
14 HTAB 39:VTAB 4:PRINT "X"
70 FOR H=38 TO 2 STEP -1
80 HTAB H:VTAB 4: PRINT "<"
90 FOR T=1 TO 100 : NEXT T
100 HTAB H:VTAB 4 : PRINT "□"
110 NEXT H : GO TO 20

יש להוסיף:

תשובה 43 מעמוד 47

65 PRINT "{CTRL-G}" יש להוסיף או לשנות:
110 NEXT H:PRINT "{CTRL-G}":GO TO 20

תשובה 44 מעמוד 47

```
10 HOME
20 FOR H=15 TO 30 STEP 5
30 HTAB H:VTAB 8:PRINT "*"
40 HTAB H:VTAB 9:PRINT "I"
50 HTAB H:VTAB 10:PRINT "I"
60 HTAB H:VTAB 11:PRINT "###"
70 NEXT H
80 FOR H=1 TO 35
90 HTAB H:VTAB 8:PRINT ">"
100 FOR T=1 TO 100 : NEXT T
110 HTAB H:VTAB 8:PRINT "□"
120 NEXT H
```

תשובה 45 מעמוד 49

```
10 HOME
20 FOR H=10 TO 17
30 HTAB H:VTAB 10:PRINT "*"
40 HTAB H:VTAB 14:PRINT "*"
50 NEXT H
60 FOR V=10 TO 14
70 HTAB 10:VTAB V:PRINT "*"
80 HTAB 17:VTAB V:PRINT "*"
90 NEXT V
```

תשובה 46 מעמוד 50

15 INPUT L יש להוסיף או לשנות:
20 FOR H=10 TO 9+L
80 HTAB 9+L : VTAB V:PRINT "*"

תשובה 52 מעמוד 54

יש להוסיף:

```
80 FOR H=B TO A STEP -1
90 HTAB H:VTAB D:PRINT " □ "
100 NEXT H
```

תשובה 53 מעמוד 55

יש להוסיף:

```
15 FOR N=1 TO 3
110 FOR V=D TO C STEP -1
120 HTAB A:VTAB V:PRINT " □ "
130 NEXT V
140 A=A-2 : B=B+2 : C=C-2 : D=D+2
150 NEXT N
```


עד כה יצאו סדרות "מחשבת" עבור:

● מחשב 64 COMMODORE

(3 החוברות הראשונות מתאימות גם ל-VIC-20)

ZX-SPECTRUM

● מחשבי SINCLAIR:

ZX-81

● מחשבי APPLE

● מחשבי ATARI

● מחשב SPECTRVIDEO

● מחשב AMSTRAD

● מחשב IBM-PC

● מחשב 128 COMMODORE

יצא לאור "שעשועי מחשבת-1"

"שעשועי מחשבת-1" – מציינת את

תחילתה של סדרה חדשה של

"מחשבת" ומיועדת לכל חובבי

המחשבים הביתיים המחפשים אתגרים
תכנותיים.

מה תמצא בחוברת הראשונה של

"שעשועי מחשבת-1"?

הצעות מפורטות לפיתוח 5 משחקי

מחשב המתאימים מאד לתכנות בשפת

BASIC.

החוברת אינה כוללת אף תוכנית – את

זאת אנו משאירים לך לפתח.

בנוסף לכך תוכל להשתתף בהגרלה

נושאת פרסים של "מחשבת-BUG".

(פרטים תמצא בחוברת עצמה).



BUS
290/12.70
12.00



ביחידת לימוד זאת הנקראת
"שעונים דיגיטליים ופירמידות"
תלמד להשתמש בכמה מהמושגים החשובים
ביותר בשפת ה-BASIC:
• משתנים מספריים ומחרוזתיים.
• לולאות סופיות (FOR-NEXT).
תוך כדי מסענו נשתמש על כלים תכנותיים נוספים
וכל זאת תוך לימוד וניצול תכונותיו המיוחדות
של מחשב ה-APPLE.
נראה כיצד מושגים חדשים אלו יפתחו בפניך עולם
ומלאן: החל בבניית שעון דיגיטלי וכלה בבניית "פירמידות".
יחידת לימוד זאת בנויה כך, שהיא תהווה עבורך
אתגר מחשבתי ותכנותי גם אם אתה כבר יודע BASIC.

DESIGN GROUP

מחשבת מ.ל. בע"מ – מערכות למידה